
Introduction
    

Broadway is a library for building concurrent and multi-stage data ingestion and data processing pipelines with Elixir. Broadway pipelines are concurrent and robust, thanks to the Erlang VM and its actors. It features:
	Back-pressure
	Automatic acknowledgements at the end of the pipeline
	Batching
	Fault tolerance
	Graceful shutdown
	Built-in testing
	Custom failure handling
	Ordering and partitioning
	Rate-limiting
	Metrics


  
    
  
  Official Producers


Currently we officially support four Broadway producers:
	Amazon SQS: Source - Guide
	Apache Kafka: Source - Guide
	Google Cloud Pub/Sub: Source - Guide
	RabbitMQ: Source - Guide

The guides links above will help you get started with your adapter of choice. For API reference, you can check out the Broadway module.

  
    
  
  Non-official (Off-Broadway) Producers


For those interested in rolling their own Broadway Producers (which we actively encourage!), we recommend using the OffBroadway namespace, mirroring the Off-Broadway theaters. For example, if you want to publish your own integration with Amazon SQS, you can package it as off_broadway_sqs, which uses the OffBroadway.SQS namespace.
The following Off-Broadway libraries are available (feel free to send a PR adding your own in alphabetical order):
	off_broadway_amqp10: Guide
	off_broadway_kafka: Guide
	off_broadway_memory: Guide
	off_broadway_redis: Guide
	off_broadway_redis_stream: Guide
	off_broadway_splunk: Guide



  

    
Amazon SQS
    

Amazon Simple Queue Service (SQS) is a highly scalable distributed message
queuing service provided by Amazon.com. AWS SQS offers two types of message
queues:
	Standard
	Nearly unlimited throughput
	Best-effort ordering
	At-least-once delivery


	FIFO
	Limited number of transactions per second (TPS).
See Amazon SQS FIFO
developer guide for more information on limits.
	Order in which messages are sent/received is strictly preserved
	Exactly-once delivery



Broadway can work seamlessly with both, Standard and FIFO queues.

  
    
  
  Getting Started


In order to use Broadway with SQS, we need to:
	Create a SQS queue (or use an existing one)
	Configure our Elixir project to use Broadway
	Define your pipeline configuration
	Implement Broadway callbacks
	Run the Broadway pipeline
	Tuning the configuration (Optional)


  
    
  
  Create a SQS queue


Amazon provides a comprehensive Step-by-step Guide
on creating SQS queues. In case you don't have an AWS account and want to
test Broadway locally, use can easily install ElasticMQ,
which is a message queue system that offers a SQS-compatible query interface.

  
    
  
  Configure the project


In this guide we're going to use BroadwaySQS,
which is a Broadway SQS Connector provided by Dashbit.

  
    
  
  Starting a new project


If you plan to start a new project, just run:
$ mix new my_app --sup

The --sup flag instructs Elixir to generate an application with a supervision tree.

  
    
  
  Setting up dependencies


Add :broadway_sqs to the list of dependencies in mix.exs along the HTTP
client of your choice (defaults to :hackney):
def deps do
  [
    ...
    {:broadway_sqs, "~> 0.7"},
    {:hackney, "~> 1.9"},
  ]
end
Don't forget to check for the latest version of dependencies.

  
    
  
  Define the pipeline configuration


Broadway is a process-based behaviour and to define a Broadway
pipeline, we need to define three functions: start_link/1,
handle_message/3 and handle_batch/4. We will cover start_link/1
in this section and the handle_ callbacks in the next one.
Similar to other process-based behaviour, start_link/1 simply
delegates to Broadway.start_link/2, which should define the
producers, processors, and batchers in the Broadway pipeline.
Assuming we want to consume messages from a queue called
my_queue, the minimal configuration would be:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  def start_link(_opts) do
    Broadway.start_link(__MODULE__,
      name: __MODULE__,
      producer: [
        module: {BroadwaySQS.Producer,
                 queue_url: "https://us-east-2.queue.amazonaws.com/100000000001/my_queue"}
      ],
      processors: [
        default: []
      ],
      batchers: [
        default: [
          batch_size: 10,
          batch_timeout: 2000
        ]
      ]
    )
  end

  ...callbacks...
end
The above configuration also assumes that you have the AWS credentials
set up in your environment, for instance, by having the AWS_ACCESS_KEY_ID
and AWS_SECRET_ACCESS_KEY environment variables set. If that's
not the case, you will need to pass that information to the client so it
can properly connect to the AWS servers. Here is how you can do it:
...
producer: [
  module:
    {BroadwaySQS.Producer,
     queue_url: "https://us-east-2.queue.amazonaws.com/100000000001/my_queue",
     config: [
       access_key_id: "YOUR_AWS_ACCESS_KEY_ID",
       secret_access_key: "YOUR_AWS_SECRET_ACCESS_KEY"
     ]}
]
...
For a full list of options for BroadwaySQS.Producer, please see
BroadwaySQS documentation.
For general information about setting up Broadway, see Broadway
module docs as well as Broadway.start_link/2.
Note: Even though batching is optional since Broadway v0.2, we recommend that all Amazon SQS
pipelines have at least a default batcher. This lets you control the exact batch
size and frequency that messages are acknowledged to Amazon SQS, often leading to
pipelines that are more cost and time efficient.


  
    
  
  Implement Broadway callbacks


In order to process incoming messages, we need to implement the
required callbacks. For the sake of simplicity, we're considering that
all messages received from the queue are just numbers:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  ...start_link...

  @impl true
  def handle_message(_, %Message{data: data} = message, _) do
    message
    |> Message.update_data(fn data -> data * data end)
  end

  @impl true
  def handle_batch(_, messages, _, _) do
    list = messages |> Enum.map(fn e -> e.data end)
    IO.inspect(list, label: "Got batch of finished jobs from processors, sending ACKs to SQS as a batch.")
    messages
  end
end
We are not doing anything fancy here, but it should be enough for our
purpose. First we update the message's data individually inside
handle_message/3 and then we print each batch inside handle_batch/4.
For more information, see Broadway.handle_message/3 and
Broadway.handle_batch/4.

  
    
  
  Run the Broadway pipeline


To run your Broadway pipeline, you just need to add as a child in
a supervision tree. Most applications have a supervision tree defined
at lib/my_app/application.ex. You can add Broadway as a child to a
supervisor as follows:
children = [
  {MyBroadway, []}
]

Supervisor.start_link(children, strategy: :one_for_one)
Now the Broadway pipeline should be started when your application starts.
Also, if your Broadway has any dependency (for example, it needs to talk
to the database), make sure that Broadway is listed after its dependencies
in the supervision tree.

  
    
  
  Tuning the configuration


Some of the configuration options available for Broadway come already with a
"reasonable" default value. However those values might not suit your
requirements. Depending on the number of messages you get, how much processing
they need and how much IO work is going to take place, you might need completely
different values to optimize the flow of your pipeline. The concurrency option
available for every set of producers, processors and batchers, among with
max_demand, batch_size, and batch_timeout can give you a great deal
of flexibility.
The concurrency option controls the concurrency level in each layer of
the pipeline.
See the notes on Producer concurrency
and Batcher concurrency
for details.
Here's an example on how you could tune them according to
your needs.
defmodule MyBroadway do
  use Broadway

  def start_link(_opts) do
    Broadway.start_link(__MODULE__,
      name: __MODULE__,
      producer: [
        ...
        concurrency: 10,
      ],
      processors: [
        default: [
          concurrency: 100,
          max_demand: 1,
        ]
      ],
      batchers: [
        default: [
          batch_size: 10,
          concurrency: 10,
        ]
      ]
    )
  end

  ...callbacks...
end
In order to get a good set of configurations for your pipeline, it's
important to respect the limitations of the servers you're running,
as well as the limitations of the services you're providing/consuming
data to/from. Broadway comes with telemetry, so you can measure your
pipeline and help ensure your changes are effective.


  

    
Apache Kafka
    

Kafka is a distributed streaming platform that has three key capabilities:
	Publish and subscribe to streams of records
	Store streams of records in a fault-tolerant durable way
	Process streams of records as they occur


  
    
  
  Getting Started


In order to use Broadway with Kafka, we need to:
	Create a stream of records (or use an existing one)
	Configure your Elixir project to use Broadway
	Define your pipeline configuration
	Implement Broadway callbacks
	Run the Broadway pipeline


  
    
  
  Create a stream of records (or use an existing one)


In case you don't have Kafka installed yet, please follow the instructions on Kafka's
Quickstart for a clean installation. After
initializing Kafka, you can create a new stream by running:
$ kafka-topics --create --zookeeper localhost:2181 --partitions 3 --topic test


  
    
  
  Configure your Elixir project to use Broadway


This guide describes the steps necessary to integrate Broadway with Kafka using
BroadwayKafka,
which is a Broadway Kafka Connector provided by Dashbit.
BroadwayKafka can subscribe to one or more topics and process streams of records
using Kafka's Consumer API.
Each GenStage producer initialized by BroadwayKafka will be available as a consumer,
all registered using the same self-labeled consumer group. Each record published to a
topic/partition will be delivered to one consumer instance within each consumer group.
Bear in mind that a topic/partition can be assigned to any consumer instance that has
been subscribed using the same consumer group, i.e, any Broadway instance or application
running on any machine connected to the Kafka cluster.

  
    
  
  Starting a new project


Create a new project running:
$ mix new my_app --sup

The --sup flag instructs Elixir to generate an application with a supervision tree.

  
    
  
  Setting up dependencies


Add :broadway_kafka to the list of dependencies in mix.exs:
def deps do
  [
    ...
    {:broadway_kafka, "~> 0.3"}
  ]
end
Don't forget to check for the latest version of dependencies.

  
    
  
  Define the pipeline configuration


Broadway is a process-based behaviour and to define a Broadway pipeline,
we need to define three functions: start_link/1, handle_message/3
and optionally handle_batch/4. We will cover start_link/1 in this
section and the handle_ callbacks in the next one.
Similar to other process-based behaviours, start_link/1 simply
delegates to Broadway.start_link/2, which should define the
producers, processors, and batchers in the Broadway pipeline.
Assuming we want to consume messages from a topic called
test, one possible configuration would be:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  def start_link(_opts) do
    Broadway.start_link(__MODULE__,
      name: __MODULE__,
      producer: [
        module:
          {BroadwayKafka.Producer,
           [
             hosts: [localhost: 9092],
             group_id: "group_1",
             topics: ["test"]
           ]},
        concurrency: 1
      ],
      processors: [
        default: [
          concurrency: 10
        ]
      ],
      batchers: [
        default: [
          batch_size: 100,
          batch_timeout: 200,
          concurrency: 10
        ]
      ]
    )
  end

  ...callbacks...
end
Note: Pipelines built on top of BroadwayKafka are automatically partitioned.
So even though there are multiple processes (stages), these processes will preserve
Kafka's ordering semantics when it comes to topics/partitions. Internally, this is
achieved by making sure all messages from the same topic/partition will always be
forwarded to the same processor and batch processor.

For a full list of options for BroadwayKafka.Producer, refer to the
official BroadwayKafka documentation.
For general information about setting up Broadway, see Broadway
module docs as well as Broadway.start_link/2.

  
    
  
  Implement Broadway callbacks


In order to process incoming messages, we need to implement the
required callbacks. For the sake of simplicity, we're considering that
all messages received from the topic are just numbers:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  ...start_link...

  @impl true
  def handle_message(_, message, _) do
    message
    |> Message.update_data(fn data -> {data, String.to_integer(data) * 2} end)
  end

  @impl true
  def handle_batch(_, messages, _, _) do
    list = messages |> Enum.map(fn e -> e.data end)
    IO.inspect(list, label: "Got batch")
    messages
  end
end
We are not doing anything fancy here, but it should be enough for our
purpose. First, we update the message's data individually inside
handle_message/3 and then we print each batch inside handle_batch/4.
For more information, see Broadway.handle_message/3 and
Broadway.handle_batch/4.
Note: Since Broadway v0.2, batching is optional. In case you don't need to
group messages as batches for further processing/publishing, you can remove
the :batchers configuration along with the handle_batch/4 callback.


  
    
  
  Run the Broadway pipeline


To run your Broadway pipeline, you just need to add as a child in
a supervision tree. Most applications have a supervision tree defined
at lib/my_app/application.ex. You can add Broadway as a child to a
supervisor as follows:
children = [
  {MyBroadway, []}
]

Supervisor.start_link(children, strategy: :one_for_one)
Now the Broadway pipeline should be started when your application starts.
Also, if your Broadway has any dependency (for example, it needs to talk
to the database), make sure that Broadway is listed after its dependencies
in the supervision tree.
You can now test your pipeline by entering an iex session:
$ iex -S mix

If everything went fine, you should see lots of info log messages like this
one coming from the :brod supervisors:
15:14:04.356 [info]  [supervisor: {:local, :brod_sup}, started: [pid: #PID<0.251.0>, id: :test_client, mfargs: {:brod_client, :start_link, [[localhost: 9092], :test_client, []]}, restart_type: {:permanent, 10}, shutdown: 5000, child_type: :worker]]
Brod is the client that BroadwayKafka uses
under the hood to communicate with Kafka.

  
    
  
  Sending messages to Kafka


Finally, we can send some sample messages to Kafka using using :brod with the following snippet:
topic = "test"
client_id = :my_client
hosts = [localhost: 9092]

:ok = :brod.start_client(hosts, client_id, _client_config=[])
:ok = :brod.start_producer(client_id, topic, _producer_config = [])

Enum.each(1..1000, fn i ->
  partition = rem(i, 3)
  :ok = :brod.produce_sync(client_id, topic, partition, _key="", "#{i}")
end)
You should see the output showing the generated batches:
Got batch: [
  {"2", 4},
  {"5", 10},
  {"8", 16},
  {"11", 22},
  {"14", 28},
  ...
]
Got batch: [
  {"3", 6},
  {"6", 12},
  {"9", 18},
  {"12", 24},
  {"15", 30},
  ...
]

  
    
  
  Tuning the configuration


Some of the configuration options available for Broadway come already with a
"reasonable" default value. However, those values might not suit your
requirements. Depending on the number of records you get, how much processing
they need and how much IO work is going to take place, you might need completely
different values to optimize the flow of your pipeline. The concurrency option
available for every set of producers, processors and batchers, along with
batch_size and batch_timeout can give you a great deal of flexibility.
See the notes on Producer concurrency
and Batcher concurrency
for details.
By setting the concurrency option, you define the number of concurrent processes
that will be started by Broadway, allowing you to have full control over the
concurrency level in each layer of the pipeline. Keep in mind that since the
concurrency model provided by Kafka is based on partitioning, in order to take
full advantage of this model, you need to set the concurrency option for
your processors and batchers accordingly. Having less concurrency than topic/partitions
assigned will result in individual processors handling more than one partition,
decreasing the overall level of concurrency. Therefore, if you want to always be able
to process messages at maximum concurrency (assuming you have enough resources to do it),
you should increase the concurrency up front to make sure you have enough processors to
handle the extra records received from new partitions assigned.
Note: Even if you don't plan to add more partitions to a Kafka topic, your pipeline
can still receive more assignments than planned. For instance, if another consumer crashes,
the server will reassign all its topic/partition to other available consumers, including
any Broadway producer subscribed to the same topic.

There are other options that you may want to take a closer look when tuning your configuration.
The :max_bytes option, for instance, belongs to the :fetch_config group and defines the
maximum amount of data to be fetched at a time from a single partition. The default is
1048576 (1 MiB). Setting greater values can improve throughput at the cost of more
memory consumption. For more information and other fetch options, please refer to the
"Fetch config options" in the official BroadwayKafka
documentation.
Other two important options are :offset_commit_interval_seconds and :offset_commit_on_ack.
Both belong to the main configuration and they can make a huge impact on performance.
The :offset_commit_interval_seconds defines the time interval between two
OffsetCommitRequest messages. The default is 5s.
The :offset_commit_on_ack, when set to true, tells Broadway to send an
OffsetCommitRequest immediately after each acknowledgement, bypassing any
interval defined in :offset_commit_interval_seconds. Setting this option to
false can increase performance since any commit requests will start respecting
the :offset_commit_interval_seconds option. This will usually result in fewer
requests to be sent to the server. However, setting long commit intervals might
lead to a large number of duplicated records to be processed after a server
restart or connection loss. Since it is always possible that duplicate messages
will be received by consumers, make sure your logic is idempotent when consuming
records to avoid inconsistencies. Also, bear in mind that the negative
performance impact might be insignificant if you're using batchers since only
one commit request will be performed per batch. As a basic rule, always take
into account the values of batch_size and batch_timeout whenever you're
tuning :offset_commit_interval_seconds and :offset_commit_on_ack.

  
    
  
  Handling failed messages


broadway_kafka never stops the flow of the stream, i.e. it will always ack the messages
even when they fail. Unlike queue-based connectors, where you can mark a single message as failed.
In Kafka that's not possible due to its single offset per topic/partition ack strategy. If you
want to reprocess failed messages, you need to roll your own strategy. A possible way to do that
is to implement handle_failed/2 and send failed messages to a separated stream or queue for
later processing.


  

    
Google Cloud Pub/Sub
    

Cloud Pub/Sub is a fully-managed real-time messaging service provided by Google.

  
    
  
  Getting Started


In order to use Broadway with Cloud Pub/Sub you need to:
	Setup a Cloud Pub/Sub project
	Configure your Elixir project to use Broadway
	Define your pipeline configuration
	Implement Broadway callbacks
	Run the Broadway pipeline
	Tune the configuration (Optional)

If you are just getting familiar with Google Pub/Sub, refer to the documentation
to get started. Instead of testing against a live environment, you may also consider using the
emulator to simulate integrating with Cloud
Pub/Sub.
If you have an existing project, topic, subscription, and credentials, you can skip step
1 and jump to Configure the project
section.

  
    
  
  Setup Cloud Pub/Sub project


In this tutorial we'll use the gcloud command-line tool
to set everything up in Google Cloud. Alternatively, you can roughly follow this guide by using
Cloud Console.
To install gcloud follow the documentation. If you are
on macOS you may consider installing it with Homebrew:
$ brew install --cask google-cloud-sdk

Now, authenticate the CLI:
$ gcloud auth login

Then, create a new project:
$ gcloud projects create test-pubsub

A new topic:
$ gcloud pubsub topics create test-topic --project test-pubsub
Created topic [projects/test-pubsub/topics/test-topic].

Note: If you run this command immediately after creating a new Google Cloud project, you may receive an error indicating that your project's organization policy is still being provisioned. Just wait a couple minutes and try again.

And a new subscription:
$ gcloud pubsub subscriptions create test-subscription --project test-pubsub --topic test-topic
Created subscription [projects/test-pubsub/subscriptions/test-subscription].

We also need a service account, an IAM
policy, as well as API credentials in order to programmatically work with the service. First, let's
create the service account:
$ gcloud iam service-accounts create test-account --project test-pubsub
Created service account [test-account].

Then the policy. For simplicity we add the general role roles/editor, but make sure to
examine the available roles
and choose the one that best suits your use case:
$ gcloud projects add-iam-policy-binding test-pubsub \
    --member serviceAccount:test-account@test-pubsub.iam.gserviceaccount.com \
    --role roles/editor
Updated IAM policy for project [test-pubsub].
(...)

And now the credentials:
$ gcloud iam service-accounts keys create credentials.json --iam-account=test-account@test-pubsub.iam.gserviceaccount.com
created key [xxx] of type [json] as [key] for [test-account@test-pubsub.iam.gserviceaccount.com]

This command generated a credentials.json file which will be useful later. Note, the IAM account
pattern is <account>@<project>.iam.gserviceaccount.com. Run gcloud iam service-accounts list --project test-pubsub
to see all service accounts associated with the given project.
Finally, we need to enable Pub/Sub for our project:
$ gcloud services enable pubsub --project test-pubsub
Operation "operations/xxx" finished successfully.


  
    
  
  Configure the project


In this guide we're going to use BroadwayCloudPubSub,
which is a Broadway Cloud Pub/Sub Connector provided by Dashbit.

  
    
  
  Starting a new project


If you plan to start a new project, just run:
$ mix new my_app --sup

The --sup flag instructs Elixir to generate an application with a supervision tree.

  
    
  
  Setting up dependencies


Add :broadway_cloud_pub_sub to the list of dependencies in mix.exs, along with the Google
Cloud authentication library of your choice (defaults to :goth):
defp deps() do
  [
    ...
    {:broadway_cloud_pub_sub, "~> 0.7"},
    {:goth, "~> 1.0"}
  ]
end
Don't forget to check for the latest version of dependencies.

  
    
  
  Define the pipeline configuration


Broadway is a process-based behaviour and to define a Broadway pipeline, we need to define three
functions: start_link/1, handle_message/3 and handle_batch/4. We will cover start_link/1
in this section and the handle_ callbacks in the next one.
Similar to other process-based behaviour, start_link/1 simply delegates to
Broadway.start_link/2, which should define the producers, processors, and batchers in the
Broadway pipeline. Assuming we want to consume messages from the test-subscription, the minimal
configuration would be:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  def start_link(_opts) do
    Broadway.start_link(__MODULE__,
      name: __MODULE__,
      producer: [
        module:
          {BroadwayCloudPubSub.Producer,
           subscription: "projects/test-pubsub/subscriptions/test-subscription"}
      ],
      processors: [
        default: []
      ],
      batchers: [
        default: [
          batch_size: 10,
          batch_timeout: 2_000
        ]
      ]
    )
  end

  ...callbacks...
end
For a full list of options for BroadwayCloudPubSub.Producer, please see the
documentation.
For general information about setting up Broadway, see Broadway module docs as well as
Broadway.start_link/2.
Note: Even though batching is optional since Broadway v0.2, we recommend all Cloud Pub/Sub
pipelines to have at least a default batcher, as that allows you to control the exact batch
size and frequency that messages are acknowledged to Cloud Pub/Sub, which often leads to
pipelines that are more cost and time efficient.


  
    
  
  Implement Broadway callbacks


In order to process incoming messages, we need to implement the required callbacks. For the sake
of simplicity, we're considering that all messages received from the queue are strings and our
processor calls String.upcase/1 on them:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  ...start_link...

  def handle_message(_, %Message{data: data} = message, _) do
    message
    |> Message.update_data(fn data -> String.upcase(data) end)
  end

  def handle_batch(_, messages, _, _) do
    list = messages |> Enum.map(fn e -> e.data end)
    IO.inspect(list, label: "Got batch of finished jobs from processors, sending ACKs to Pub/Sub as a batch.")
    messages
  end
end
We are not doing anything fancy here, but it should be enough for our purpose. First we update the
message's data individually inside handle_message/3 and then we print each batch inside
handle_batch/4.
For more information, see Broadway.handle_message/3 and Broadway.handle_batch/4.

  
    
  
  Run the Broadway pipeline


To run your Broadway pipeline, you need to add it as a child in a supervision tree. Most
applications have a supervision tree defined at lib/my_app/application.ex. You can add Broadway
as a child to a supervisor as follows:
children = [
  {MyBroadway, []}
]

Supervisor.start_link(children, strategy: :one_for_one)
The final step is to configure credentials. You can set the following environment variable:
export GOOGLE_APPLICATION_CREDENTIALS=/path/to/credentials.json
See Goth documentation for alternative ways of authenticating
with the API.
Now the Broadway pipeline should be started when your application starts. Also, if your Broadway
pipeline has any dependency (for example, it needs to talk to the database), make sure that
it is listed after its dependencies in the supervision tree.
If you followed the previous section about setting the project with gcloud, you can now test the
the pipeline. In one terminal tab start the application:
$ iex -S mix

And in another tab, send a couple of test messages to Pub/Sub:
$ gcloud pubsub topics publish  projects/test-pubsub/topics/test-topic --message "test 1"
messageIds:
- '651428033718119'

gcloud pubsub topics publish  projects/test-pubsub/topics/test-topic --message "test 2"
messageIds:
- '651427034966696'

Now, In the first tab, you should see output similar to:
Got batch of finished jobs from processors, sending ACKs to Pub/Sub as a batch.: ["TEST 1", "TEST 2"]

  
    
  
  Tuning the configuration


Some of the configuration options available for Broadway come already with a
"reasonable" default value. However those values might not suit your
requirements. Depending on the number of messages you get, how much processing
they need and how much IO work is going to take place, you might need completely
different values to optimize the flow of your pipeline. The concurrency option
available for every set of producers, processors and batchers, among with
max_demand, batch_size, and batch_timeout can give you a great deal
of flexibility.
The concurrency option controls the concurrency level in each layer of
the pipeline.
See the notes on Producer concurrency
and Batcher concurrency
for details.
Here's an example on how you could tune them according to
your needs.
defmodule MyBroadway do
  use Broadway

  def start_link(_opts) do
    Broadway.start_link(__MODULE__,
      name: __MODULE__,
      producer: [
        ...
        concurrency: 10,
      ],
      processors: [
        default: [
          concurrency: 100,
          max_demand: 1,
        ]
      ],
      batchers: [
        default: [
          batch_size: 10,
          concurrency: 10,
        ]
      ]
    )
  end

  ...callbacks...
end
In order to get a good set of configurations for your pipeline, it's
important to respect the limitations of the servers you're running,
as well as the limitations of the services you're providing/consuming
data to/from. Broadway comes with telemetry, so you can measure your
pipeline and help ensure your changes are effective.


  

    
RabbitMQ
    

RabbitMQ is an open source message broker designed to be highly scalable and
distributed. It supports multiple protocols including the Advanced Message
Queuing Protocol (AMQP).

  
    
  
  Getting Started


In order to use Broadway with RabbitMQ, we need to:
	Create a queue (or use an existing one)
	Configure our Elixir project to use Broadway
	Define your pipeline configuration
	Implement Broadway callbacks
	Run the Broadway pipeline
	Tuning the configuration (Optional)

In case you want to work with an existing queue, you can skip step 1
and jump to Configure the project.
Note: BroadwayRabbitMQ does not automatically create any queue. If you
configure a pipeline with a non-existent queue, the producers will crash,
bringing down the pipeline.


  
    
  
  Create a queue


RabbitMQ runs on many operating systems. Please see
Downloading and Installing RabbitMQ for
further information. Also, make sure you have the
Management plugin enabled, which ships
with the command line tool, rabbitmqadmin.
After successfully installing RabbitMQ, you can declare a new queue with the
following command:
$ rabbitmqadmin declare queue name=my_queue durable=true

You can list all declared queues to see our the one we've just created:
$ rabbitmqctl list_queues
Timeout: 60.0 seconds ...
Listing queues for vhost / ...
name      messages
my_queue  0


  
    
  
  Configure the project


In this guide, we're going to use BroadwayRabbitMQ,
which is a Broadway RabbitMQ Connector provided by Dashbit.

  
    
  
  Starting a new project


If you're creating a new project, run:
$ mix new my_app --sup

The --sup flag instructs Elixir to generate an application with a supervision tree.

  
    
  
  Setting up dependencies


Add :broadway_rabbitmq to the list of dependencies in mix.exs:
def deps do
  [
    ...
    {:broadway_rabbitmq, "~> 0.7"},
  ]
end
Don't forget to check for the latest version of dependencies.

  
    
  
  Define the pipeline configuration


Broadway is a process-based behaviour and to define a Broadway pipeline,
we need to define three functions: start_link/1, handle_message/3
and optionally handle_batch/4. We will cover start_link/1 in this
section and the handle_ callbacks in the next one.
Similar to other process-based behaviours, start_link/1 simply
delegates to Broadway.start_link/2, which should define the
producers, processors, and batchers in the Broadway pipeline.
Assuming we want to consume messages from a queue called
my_queue, one possible configuration would be:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  def start_link(_opts) do
    Broadway.start_link(__MODULE__,
      name: MyBroadway,
      producer: [
        module: {BroadwayRabbitMQ.Producer,
          queue: "my_queue",
          qos: [
            prefetch_count: 50,
          ]
        },
        concurrency: 1
      ],
      processors: [
        default: [
          concurrency: 50
        ]
      ],
      batchers: [
        default: [
          batch_size: 10,
          batch_timeout: 1500,
          concurrency: 5
        ]
      ]
    )
  end

  ...callbacks...
end
If you're consuming data from an existing broker that requires authorization,
you'll need to provide your credentials using the connection option:
...
producer: [
  module: {BroadwayRabbitMQ.Producer,
    queue: "my_queue",
    connection: [
      username: "user",
      password: "password",
    ]
    ...
  }
]
...
For the full list of connection options, please see
AMQP.Connection.open/1
For general information about setting up Broadway, see Broadway
module docs as well as Broadway.start_link/2.

  
    
  
  Implement Broadway callbacks


In order to process incoming messages, we need to implement the
required callbacks. For the sake of simplicity, we're considering that
all messages received from the queue are just numbers:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  ...start_link...

  @impl true
  def handle_message(_, message, _) do
    message
    |> Message.update_data(fn data -> {data, String.to_integer(data) * 2} end)
  end

  @impl true
  def handle_batch(_, messages, _, _) do
    list = messages |> Enum.map(fn e -> e.data end)
    IO.inspect(list, label: "Got batch")
    messages
  end
end
We are not doing anything fancy here, but it should be enough for our
purpose. First, we update the message's data individually inside
handle_message/3 and then we print each batch inside handle_batch/4.
For more information, see Broadway.handle_message/3 and
Broadway.handle_batch/4.
Note: Since Broadway v0.2, batching is optional. In case you don't need to
group messages as batches for further processing/publishing, you can remove
the :batchers configuration along with the handle_batch/4 callback. This
is perfectly fine for RabbitMQ, where messages are acknowledged individually
and never as a batch.


  
    
  
  Run the Broadway pipeline


To run your Broadway pipeline, you just need to add as a child in
a supervision tree. Most applications have a supervision tree defined
at lib/my_app/application.ex. You can add Broadway as a child to a
supervisor as follows:
children = [
  {MyBroadway, []}
]

Supervisor.start_link(children, strategy: :one_for_one)
Now the Broadway pipeline should be started when your application starts.
Also, if your Broadway has any dependency (for example, it needs to talk
to the database), make sure that Broadway is listed after its dependencies
in the supervision tree.
You can now test your pipeline by entering an iex session:
$ iex -S mix

If everything went fine, you should see lots of info log messages from the amqp
supervisors. If you think that's too verbose and want to do something
about it, please take a look at the "Log related to amqp supervisors are too verbose"
subsection in the amqp's  Troubleshooting
documentation.
Finally, let's generate some sample messages to be consumed by Broadway with the
following code:
{:ok, connection} = AMQP.Connection.open
{:ok, channel} = AMQP.Channel.open(connection)
AMQP.Queue.declare(channel, "my_queue", durable: true)

Enum.each(1..5000, fn i ->
  AMQP.Basic.publish(channel, "", "my_queue", "#{i}")
end)
AMQP.Connection.close(connection)
You should see the output showing the generated batches:
Got batch: [
  {"7", 14},
  {"5", 10},
  {"8", 16},
  {"98", 196},
  {"6", 12},
  {"97", 194},
  {"9", 18},
  {"99", 198},
  {"10", 20},
  {"100", 200}
]
Got batch: [
  {"29", 58},
  {"32", 64},
  ...
]

  
    
  
  Tuning the configuration


Some of the configuration options available for Broadway come already with a
"reasonable" default value. However, those values might not suit your
requirements. Depending on the number of messages you get, how much processing
they need and how much IO work is going to take place, you might need completely
different values to optimize the flow of your pipeline. The concurrency option
available for every set of producers, processors and batchers, among with
max_demand, batch_size, and batch_timeout can give you a great deal
of flexibility. The concurrency option controls the concurrency level in
each layer of the pipeline.
See the notes on Producer concurrency
and Batcher concurrency
for details.
Another important option to take into account is the :prefetch_count.
RabbitMQ will continually push new messages to Broadway as it receives them.
The :prefetch_count setting provides back-pressure by instructing RabbitMQ to limit the number of unacknowledged messages a consumer will have at a given moment.
See the "Back-pressure and :prefetch_count"
section of the BroadwayRabbitMQ documentation for details.
In order to get a good set of configurations for your pipeline, it's
important to respect the limitations of the servers you're running,
as well as the limitations of the services you're providing/consuming
data to/from. Broadway comes with telemetry, so you can measure your
pipeline and help ensure your changes are effective.


  

    
Custom Producers
    

If you want to use Broadway but there is no existing Broadway producer
for the technology of your choice, you can integrate any existing GenStage
producer into the pipeline with relative ease.

  
    
  
  Example


In general, producers must generate %Broadway.Message{} structs in order
to be processed by Broadway. In case you need to use an existing GenStage
producer and you don't want to change its original implementation,
you'll have to set the producer's :transformer option to translate the
generated events into Broadway messages.
In the following example the producer is a regular GenStage, i.e., it
produces plain events that cannot be processed by Broadway directly:
defmodule Counter do
  use GenStage

  def start_link(number) do
    GenStage.start_link(Counter, number)
  end

  def init(counter) do
    {:producer, counter}
  end

  def handle_demand(demand, counter) when demand > 0 do
    events = Enum.to_list(counter..counter+demand-1)
    {:noreply, events, counter + demand}
  end
end
By using a transformer, you can tell Broadway to transform all events
generated by the producer into proper Broadway messages:
defmodule MyBroadway do
  use Broadway

  alias Broadway.Message

  def start_link(_opts) do
    Broadway.start_link(__MODULE__,
      name: __MODULE__,
      producer: [
        module: {Counter, 1},
        transformer: {__MODULE__, :transform, []}
      ],
      processors: [
        default: [concurrency: 10]
      ],
      batchers: [
        default: [concurrency: 2, batch_size: 5],
      ]
    )
  end

  ...callbacks...

  def transform(event, _opts) do
    %Message{
      data: event,
      acknowledger: {__MODULE__, :ack_id, :ack_data}
    }
  end

  def ack(:ack_id, successful, failed) do
    # Write ack code here
  end
end
Notice that you need to pass two options to the producer:
	:module - a tuple representing the GenStage producer as {mod, arg}.
Where mod is module that implements the GenStage behaviour and arg
the argument that will be given to the init callback of the GenStage.
It is very important to note that Broadway will not call the
child_spec/1 or start_link/1 function of the producer. That's
because Broadway wraps the producer to augment it with extra features.

	:transformer - a module-function-args tuple that will be invoked
inside the producer, for every producer message, that should create
a Broadway.Message struct with the data and acknowledger fields.


See the Broadway.Acknowledger module for more information on defining
and setting up acknowledgements.


  

    
Architecture
    

Broadway's architecture is built on top of GenStage. That means we structure
our processing units as independent stages that are responsible for one
individual task in the pipeline. By implementing the Broadway behaviour,
we define a GenServer process that wraps a Supervisor to manage and
own our pipeline.

  
    
  
  The pipeline model


                                       [producers]   <- pulls data from SQS, RabbitMQ, etc.
                                            |
                                            |   (demand dispatcher)
                                            |
   handle_message/3 and  ---------->   [processors]
   prepare_messages/2 run here             / \
                                          /   \   (partition dispatcher)
                                         /     \
                                   [batcher]   [batcher]   <- one for each batcher key
                                       |           |
                                       |           |   (demand dispatcher)
                                       |           |
handle_batch/4 runs here -> [batch processor][batch processor]

  
    
  
  Internal stages


	Broadway.Producer - A wrapper around the actual producer defined by
the user. It serves as the source of the pipeline.
	Broadway.Processor - This is where messages are processed, e.g. do
calculations, convert data into a custom json format etc. Here is where
the code from handle_message/3 runs.
	Broadway.Batcher - Creates batches of messages based on the
batcher's key. One batcher for each key will be created.
	Broadway.BatchProcessor - This is where the code from handle_batch/4 runs.


  
    
  
  The supervision tree


Broadway was designed to always go back to a working state in case
of failures thanks to the use of supervisors. Our supervision tree
is designed as follows:
                                   [Broadway GenServer]
                                            |
                                            |
                                            |
                              [Broadway Pipeline Supervisor]
                             /    /   (:rest_for_one)   \    \
                           /     |                       |      \
                         /       |                       |         \
                       /         |                       |            \
                     /           |                       |               \
                   /             |                       |                  \
  [ProducerSupervisor]  [ProcessorSupervisor]   [BatchersSupervisor]    [Terminator]
    (:one_for_one)        (:one_for_all)           (:one_for_one)
         / \                    / \                /            \
        /   \                  /   \              /              \
       /     \                /     \            /                \
      /       \              /       \          /                  \
[Producer_1]  ...    [Processor_1]  ...  [BatcherSupervisor_1]     ...
                                            (:rest_for_one)
                                            /             \
                                           /               \
                                          /                 \
                                     [Batcher]   [BatchProcessorSupervisor]
                                                       (:one_for_all)
                                                        /           \
                                                       /             \
                                                      /               \
                                            [BatchProcessor_1]        ...
Both ProcessorSupervisor and BatchProcessorSupervisor are set with
max_restarts to 0. The idea is that if any process fails, we want
to restart the rest of the tree. Since Broadway callbacks are
stateless, we can handle errors and provide reports without crashing
processes. This means that the supervision tree will only shutdown
in case of unforeseen errors in Broadway's implementation.
The only exception are the producers, which contain external code
and are expected to fail. If a producer crashes, it will be restarted
by its supervisor without cascading failures until its max restarts
is reached. Broadway automatically handles those failures by making
processors automatically resubscribe to producers in case of crashes.

  
    
  
  Graceful shutdowns


The cascading failures aspect also provides safe semantics for graceful
shutdown. We know that either all processes are running OR they are all
being shutdown. Therefore, to gracefully shutdown the supervision tree,
a terminator process is activated, which starts the following steps:
	It notifies the first layer of processors that they should not
resubscribe to producers once they exit

	It tells all producers to no longer accept demand, flush all
current events, and then shutdown

	It then monitors and waits for a confirmation message from batch
processors. At this point, the terminator is effectively blocking
the supervisor until all events have been processed


This triggers a cascade effect where processors notice all of its producers
have been cancelled, causing them to flush their own events and cancels the
stages downstream, and so on and so on. This happens until batch processors
notice all of their producers have been cancelled, effectively notifying the
terminator to shutdown, allowing the outer most supervisor to go on and fully
terminate all stages, which at this point have flushed all events.


  

    
Broadway behaviour
    



      
Broadway is a concurrent, multi-stage tool for building
data ingestion and data processing pipelines.
It allows developers to consume data efficiently from different
sources, such as Amazon SQS, Apache Kafka, Google Cloud PubSub,
RabbitMQ and others.

  
    
  
  Built-in features


	Back-pressure - by relying on GenStage, we only get the amount
of events necessary from upstream sources, never flooding the
pipeline.

	Automatic acknowledgements - Broadway automatically acknowledges
messages at the end of the pipeline or in case of errors.

	Batching - Broadway provides built-in batching, allowing you to
group messages either by size and/or by time. This is important
in systems such as Amazon SQS, where batching is the most efficient
way to consume messages, both in terms of time and cost.

	Fault tolerance - Broadway pipelines are carefully designed to manage
failures. Producers are isolated from the rest of the pipeline and
automatically resubscribe to your data source in case of crashes.
At the same time, all of your Broadway callbacks are stateless, which
allows Broadway to handle any errors locally. This provides a stable
foundation that play well with your producers, regardless if their
delivery guarantees are at least once, at most once, or exactly once.

	Graceful shutdown - Broadway integrates with the VM to provide graceful
shutdown. By starting Broadway as part of your supervision tree, it will
guarantee all events are flushed once the VM shuts down.

	Built-in testing - Broadway ships with a built-in test API, making it
easy to push test messages through the pipeline and making sure the
event was properly processed.

	Custom failure handling - Broadway provides a handle_failed/2 callback
where developers can outline custom code to handle errors. For example,
if they want to move messages to another queue for further processing.

	Dynamic batching - Broadway allows developers to batch messages based on
custom criteria. For example, if your pipeline needs to build
batches based on the user_id, email address, etc, it can be done
by calling Broadway.Message.put_batch_key/2.

	Ordering and Partitioning - Broadway allows developers to partition
messages across workers, guaranteeing messages within the same partition
are processed in order. For example, if you want to guarantee all events
tied to a given user_id are processed in order and not concurrently,
you can set the :partition_by option. See "Ordering and partitioning".

	Rate limiting - Broadway allows developers to rate limit all producers in
a single node by a given number of messages in a time period, allowing
developers to easily work sources or sinks that cannot cope with a high
number of requests. See the :rate_limiting option for producers in
start_link/2.

	Metrics - Broadway uses the :telemetry library for instrumentation,
see "Telemetry" section below for more information.



  
    
  
  The Broadway behaviour


In order to use Broadway, you need to:
	Define your pipeline configuration
	Define a module implementing the Broadway behaviour


  
    
  
  Example


Broadway is a process-based behaviour, and you begin by
defining a module that invokes use Broadway. Processes
defined by these modules will often be started by a
supervisor, and so a start_link/1 function is frequently
also defined but not strictly necessary.
defmodule MyBroadway do
  use Broadway

  def start_link(_opts) do
    Broadway.start_link(MyBroadway,
      name: MyBroadwayExample,
      producer: [
        module: {Counter, []},
        concurrency: 1
      ],
      processors: [
        default: [concurrency: 2]
      ]
    )
  end

  ...callbacks...
end
Then add your Broadway pipeline to your supervision tree
(usually in lib/my_app/application.ex):
children = [
  {MyBroadway, []}
]

Supervisor.start_link(children, strategy: :one_for_one)
Adding your pipeline to your supervision tree in this way
calls the default child_spec/1 function that is generated
when use Broadway is invoked. If you would like to customize
the child spec passed to the supervisor, you can override the
child_spec/1 function in your module or explicitly pass a
child spec to the supervisor when adding it to your supervision tree.
The configuration above defines a pipeline with:
	One producer
	Two processors

Here is how this pipeline would be represented:
                     [producer_1]
                         / \
                        /   \
                       /     \
                      /       \
             [processor_1] [processor_2]   <- process each message
After the pipeline is defined, you need to implement the handle_message/3
callback which will be invoked by processors for each message.
handle_message/3 receives every message as a Broadway.Message
struct and it must return an updated message.

  
    
  
  Batching


Depending on the scenario, you may want to group processed messages as
batches before publishing your data. This is common and especially
important when working with services like AWS S3 and SQS that provide a
specific API for sending and retrieving batches. This can drastically
increase throughput and consequently improve the overall performance of
your pipeline.
To create batches, define the :batchers configuration option:
defmodule MyBroadway do
  use Broadway

  def start_link(_opts) do
    Broadway.start_link(MyBroadway,
      name: MyBroadwayExample,
      producer: [
        module: {Counter, []},
        concurrency: 1
      ],
      processors: [
        default: [concurrency: 2]
      ],
      batchers: [
        sqs: [concurrency: 2, batch_size: 10],
        s3: [concurrency: 1, batch_size: 10]
      ]
    )
  end

  # ...callbacks...
end
The configuration above defines a pipeline with:
	One producer
	Two processors
	One batcher named :sqs with two batch processors
	One batcher named :s3 with one batch processor

Here is how this pipeline would be represented:
                     [producer_1]
                         / \
                        /   \
                       /     \
                      /       \
             [processor_1] [processor_2]   <- process each message
                      /\     /\
                     /  \   /  \
                    /    \ /    \
                   /      x      \
                  /      / \      \
                 /      /   \      \
                /      /     \      \
           [batcher_sqs]    [batcher_s3]
                /\                  \
               /  \                  \
              /    \                  \
             /      \                  \
   [batch_sqs_1] [batch_sqs_2]    [batch_s3_1] <- process each batch
Additionally, you have to define the handle_batch/4 callback,
which batch processors invoke for each batch. You can then
call Broadway.Message.put_batcher/2 inside handle_message/3 to
control which batcher the message should go to.
The batcher receives processed messages and creates batches
specified by the batch_size and batch_timeout configuration. The
goal is to create a batch with at most batch_size entries within
batch_timeout milliseconds. Each message goes into a particular batch,
controlled by calling Broadway.Message.put_batch_key/2 in
handle_message/3. Once a batch is created in the batcher, it is sent
to a separate process (the batch processor) that will call handle_batch/4,
passing the batcher, the batch itself (a list of messages), a Broadway.BatchInfo
struct, and the Broadway context.
For example, imagine your producer generates integers as data.
You want to route the odd integers to SQS and the even ones to
S3. Your pipeline would look like this:
defmodule MyBroadway do
  use Broadway
  import Integer

  alias Broadway.Message

  # ...start_link...

  @impl true
  def handle_message(_, %Message{data: data} = message, _) when is_odd(data) do
    message
    |> Message.update_data(&process_data/1)
    |> Message.put_batcher(:sqs)
  end

  def handle_message(_, %Message{data: data} = message, _) when is_even(data) do
    message
    |> Message.update_data(&process_data/1)
    |> Message.put_batcher(:s3)
  end

  defp process_data(data) do
    # Do some calculations, generate a JSON representation, etc.
  end

  @impl true
  def handle_batch(:sqs, messages, _batch_info, _context) do
    # Send batch of successful messages as ACKs to SQS
    # This tells SQS that this list of messages were successfully processed
  end

  def handle_batch(:s3, messages, _batch_info, _context) do
    # Send batch of messages to S3
  end
end
See the callbacks documentation for more information on the
arguments given to each callback and their expected return types.

  
    
  
  The default batcher


Once you define the :batchers configuration key for your Broadway pipeline,
then all messages get batched. By default, unless you call
Broadway.Message.put_batcher/2, messages have their batcher set to the
:default batcher. If you don't define configuration for it, Broadway is going
to raise an error.
For example, imagine you want to batch "special" messages and handle them differently
then all other messages. You can configure your pipeline like this:
defmodule MyBroadway do
  use Broadway

  def start_link(_opts) do
    Broadway.start_link(MyBroadway,
      name: MyBroadwayExample,
      producer: [
        module: {Counter, []},
        concurrency: 1
      ],
      processors: [
        default: [concurrency: 2]
      ],
      batchers: [
        special: [concurrency: 2, batch_size: 10],
        default: [concurrency: 1, batch_size: 10]
      ]
    )
  end

  def handle_message(_, message, _) do
    if special?(message) do
      message
      |> Broadway.Message.put_batcher(:special)
    else
      message
    end
  end

  def handle_batch(:special, messages, _batch_info, _context) do
    # Handle special batch
  end

  def handle_batch(:default, messages, _batch_info, _context) do
    # Handle all other messages in batches
  end
Now you are ready to get started. See the start_link/2 function
for a complete reference on the arguments and options allowed.
Also makes sure to check out GUIDES in the documentation sidebar
for more examples, how tos and more.

  
    
  
  Acknowledgements and failures


At the end of the pipeline, messages are automatically acknowledged.
If there are no batchers, the acknowledgement will be done by processors.
The number of messages acknowledged, assuming the pipeline is running
at full scale, will be max_demand - min_demand. Since the default values
are 10 and 5 respectively, we will be acknowledging in groups of 5.
If there are batchers, the acknowledgement is done by the batchers,
using the batch_size.
In case of failures, Broadway does its best to keep the failures
contained and avoid losing messages. The failed message or batch is
acknowledged as failed immediately. For every failure, a log report
is also emitted. If your Broadway module also defines the
handle_failed/2 callback, that callback will be invoked with
all the failed messages before they get acknowledged.
Note however, that Broadway does not provide any sort of retries
out of the box. This is left completely as a responsibility of the
producer. For instance, if you are using Amazon SQS, the default
behaviour is to retry unacknowledged messages after a user-defined
timeout. If you don't want unacknowledged messages to be retried,
is your responsibility to configure a dead-letter queue as target
for those messages.

  
    
  
  Producer concurrency


Setting producer concurrency is a tradeoff between latency and internal
queueing.
For efficiency, you should generally limit the amount of internal queueing.
Whenever additional messages are sitting in a busy processor's mailbox, they
can't be delivered to another processor which may be available or become
available first.
One possible cause of internal queueing is multiple producers. This is because
each processor's demand will be sent to all producers. For example, if a
processor demands 2 messages and there are 2 producers, each producer
will try to produce 2 messages (for example, by pulling from a queue or
whatever the specific producer does) and give them to the processor. So the
processor may receive max_demand * <producer concurrency> messages.
Setting producer concurrency: 1 will reduce internal queueing. This is
likely a good choice for producers which take minimal time to produce a
message, such as BroadwayRabbitMQ, which receives messages as they are
pushed by RabbitMQ and can specify how many to prefetch.
On the other hand, when using a producer such as BroadwaySQS which must
make a network round trip to fetch from an external source, it may be better
to use multiple producers and accept some internal queueing to avoid having
fetch messages whenever there is new demand.
Measure your system to decide which setting is most appropriate.
Adding another single-producer pipeline, or another node running the
pipeline, are other ways you may consider to increase throughput.

  
    
  
  Batcher concurrency


If a batcher's concurrency is greater than 1, Broadway will use as few of
the batcher processes as possible at any given moment, attempting to satisfy
the batch_size of one batcher process within the batch_timeout before
sending messages to another.

  
    
  
  Testing


Many producers receive data from external systems and hitting the network
is usually undesirable when running the tests.
For testing purposes, we recommend developers to use Broadway.DummyProducer.
This producer does not produce any messages by itself and instead the
test_message/3 and test_batch/3 functions should be used to publish
messages.
With test_message/3, you can push a message into the pipeline and receive
a process message when the pipeline acknowledges the data you have pushed
has been processed.
Let's see an example. Imagine the following Broadway module:
defmodule MyBroadway do
  use Broadway

  def start_link() do
    producer_module = Application.fetch_env!(:my_app, :producer_module)
    producer_options = Application.get_env(:my_app, :producer_options, [])

    Broadway.start_link(__MODULE__,
      name: __MODULE__,
      producer: [
        module: {producer_module, producer_options}
      ],
      processors: [
        default: []
      ],
      batchers: [
        default: [batch_size: 10]
      ]
    )
  end

  @impl true
  def handle_message(_processor, message, _context) do
    message
  end

  @impl true
  def handle_batch(_batcher, messages, _batch_info, _context) do
    messages
  end
end
Now in config/test.exs you could do:
config :my_app,
  producer_module: Broadway.DummyProducer,
  producer_options: [] # change if required for your dev/prod producer
And we can test it like this:
defmodule MyBroadwayTest do
  use ExUnit.Case, async: true

  test "test message" do
    ref = Broadway.test_message(MyBroadway, 1)
    assert_receive {:ack, ^ref, [%{data: 1}], []}
  end
end
Note that at the end we received a message in the format of:
{:ack, ^ref, successful_messages, failure_messages}
You can use the acknowledgment to guarantee the message has been
processed and therefore any side-effect from the pipeline should be
visible.
When using test_message/3, the message will be delivered as soon as
possible, without waiting for the pipeline batch_size to be reached
or without waiting for batch_timeout. This behaviour is useful to test
and verify single messages, without imposing high timeouts to our test
suites.
In case you want to test multiple messages, then you need to use
test_batch/3. test_batch/3 will respect the batching configuration,
which most likely means you need to increase your test timeouts:
test "batch messages" do
  ref = Broadway.test_batch(MyBroadway, [1, 2, 3])
  assert_receive {:ack, ^ref, [%{data: 1}, %{data: 2}, %{data: 3}], []}, 1000
end
However, keep in mind that, generally speaking, there is no guarantee
the messages will arrive in the same order that you have sent them,
especially for large batches, as Broadway will process large batches
concurrently and order will be lost.
If you want to send more than one test message at once, then we recommend
setting the :batch_mode to :bulk, especially if you want to assert how
the code will behave with large batches. Otherwise the batcher will flush
messages as soon as possible and in small batches.
However, keep in mind that, regardless of the :batch_mode you cannot
rely on ordering, as Broadway pipelines are inherently concurrent. For
example, if you send those messages:
test "multiple batch messages" do
  ref = Broadway.test_batch(MyBroadway, [1, 2, 3, 4, 5, 6, 7], batch_mode: :bulk)
  assert_receive {:ack, ^ref, [%{data: 1}], []}, 1000
end

  
    
  
  Testing with Ecto


If you are using Ecto in your Broadway processors and you want
to run your tests concurrently, you need to tell Broadway to
use the Ecto SQL Sandbox during tests. This can be done in two
steps.
First, when you call test_messages/3 in your tests, include
the :ecto_sandbox process in the message metadata:
Broadway.test_message(MyApp.Pipeline, message, metadata: %{ecto_sandbox: self()})
Now we can use Broadway telemetry callbacks to fetch the sandbox
process and enable it inside the processor. Add to your
test/test_helper.exs:
defmodule BroadwayEctoSandbox do
  def attach(repo) do
    events = [
      [:broadway, :processor, :start],
      [:broadway, :batch_processor, :start],
    ]

    :telemetry.attach_many({__MODULE__, repo}, events, &__MODULE__.handle_event/4, %{repo: repo})
  end

  def handle_event(_event_name, _event_measurement, %{messages: messages}, %{repo: repo}) do
    with [%Broadway.Message{metadata: %{ecto_sandbox: pid}} | _] <- messages do
      Ecto.Adapters.SQL.Sandbox.allow(repo, pid, self())
    end

    :ok
  end
end

BroadwayEctoSandbox.attach(MyApp.Repo)
And now you should have concurrent Broadway tests that talk to the database.

  
    
  
  Ordering and partitioning


By default, Broadway processes all messages and batches concurrently,
which means ordering is not guaranteed. Some producers may impose some
ordering (for instance, Apache Kafka), but if the ordering comes from a
business requirement, you will have to impose the ordering yourself.
This can be done with the :partition_by option, which enforces that
messages with a given property are always forwarded to the same stage.
In order to provide partitioning throughout the whole pipeline, just
set :partition_by at the root of your configuration:
defmodule MyBroadway do
  use Broadway

  def start_link(_opts) do
    Broadway.start_link(__MODULE__,
      name: __MODULE__,
      producer: [
        module: {Counter, []},
        concurrency: 1
      ],
      processors: [
        default: [concurrency: 2]
      ],
      batchers: [
        sqs: [concurrency: 2, batch_size: 10],
        s3: [concurrency: 1, batch_size: 10]
      ],
      partition_by: &partition/1
    )
  end

  defp partition(msg) do
    msg.data.user_id
  end
In the example above, we are partitioning the pipeline by user_id.
This means any message with the same user_id will be handled by
the same processor and batch processor.
The partition function must return a non-negative integer,
starting at zero, which is routed to a stage by using the remainder
option.
If the data you want to partition by is not an integer, you can
explicitly hash it by calling :erlang.phash2/1. However, note
that hash does not guarantee an equal distribution of events
across partitions. So some partitions may be more overloaded than
others, slowing down the whole pipeline.
In the example above, we have set the same partition for all
processors and batchers. You can also specify the :partition_by
function for each "processor" and "batcher" individually.
Even partitions
Broadway partitions assume an even distribution of partitions.
This means that, if one partition is slow, it will slow down
all order partitions. This implies two things:
	Using :partition_by with a high level of concurrency can
actually be detrimental to performance. For example, if
concurrency is set to 100, you need all 100 processors to
make progress at the same time.

	Avoid using :partition_by with a low value of min_demand.
For example, setting max_demand to 1 (which implies min_demand
of 0), means that each processor will receive a single message
and only receive further messages once all processors complete.


When partitioning, the default values for concurrency (which is
equal to the number of cores) and max_demand (which is equal to
10), are good starting points.

Error semantics
Beware of the error semantics when using partitioning.
If you require messages to be processed in order and a message
fails, the partition will continue processing messages, which
may be undesired. If your producer supports retrying, the
failed message may be retried later, out of its original order.
Those issues happen regardless of Broadway and solutions to said
problems almost always need to be addressed outside of Broadway too.


  
    
  
  Telemetry


Broadway currently exposes following Telemetry events:
	[:broadway, :topology, :init] - Dispatched when the topology for
a Broadway pipeline is initialized. The config key in the metadata
contains the configuration options that were provided to
Broadway.start_link/2.
	Measurement: %{system_time: integer}
	Metadata: %{supervisor_pid: pid, config: keyword}


	[:broadway, :processor, :start] - Dispatched by a Broadway processor
before the optional prepare_messages/2
	Measurement: %{system_time: integer}

	Metadata:
%{
  topology_name: atom,
  name: atom,
  processor_key: atom,
  index: non_neg_integer,
  messages: [Broadway.Message.t],
  telemetry_span_context: reference,
  producer: {atom, list}
}



	[:broadway, :processor, :stop] -  Dispatched by a Broadway processor
after prepare_messages/2 and after all handle_message/3 callback
has been invoked for all individual messages
	Measurement: %{duration: native_time}

	Metadata:
%{
  topology_name: atom,
  name: atom,
  processor_key: atom,
  index: non_neg_integer,
  successful_messages_to_ack: [Broadway.Message.t],
  successful_messages_to_forward: [Broadway.Message.t],
  failed_messages: [Broadway.Message.t],
  telemetry_span_context: reference,
  producer: {atom, list}
}



	[:broadway, :processor, :message, :start] - Dispatched by a Broadway processor
before your handle_message/3 callback is invoked
	Measurement: %{system_time: integer}

	Metadata:
%{
  processor_key: atom,
  topology_name: atom,
  name: atom,
  index: non_neg_integer,
  message: Broadway.Message.t,
  telemetry_span_context: reference
}



	[:broadway, :processor, :message, :stop] - Dispatched by a Broadway processor
after your handle_message/3 callback has returned
	Measurement: %{duration: native_time}

	Metadata:
%{
  processor_key: atom,
  topology_name: atom,
  name: atom,
  index: non_neg_integer,
  message: Broadway.Message.t,
  telemetry_span_context: reference
}



	[:broadway, :processor, :message, :exception] - Dispatched by a Broadway processor
if your handle_message/3 callback encounters an exception
	Measurement: %{duration: native_time}

	Metadata:
%{
  processor_key: atom,
  topology_name: atom,
  name: atom,
  index: non_neg_integer,
  message: Broadway.Message.t,
  kind: kind,
  reason: reason,
  stacktrace: stacktrace,
  telemetry_span_context: reference
}



	[:broadway, :batch_processor, :start] - Dispatched by a Broadway batch processor
before your handle_batch/4 callback is invoked
	Measurement: %{system_time: integer}

	Metadata:
%{
  topology_name: atom,
  name: atom,
  index: non_neg_integer,
  messages: [Broadway.Message.t],
  batch_info: Broadway.BatchInfo.t,
  telemetry_span_context: reference,
  producer: {atom, list}
}



	[:broadway, :batch_processor, :stop] - Dispatched by a Broadway batch
processor after your handle_batch/4 callback has returned
	Measurement: %{duration: native_time}

	Metadata:
%{
  topology_name: atom,
  name: atom,
  index: non_neg_integer,
  successful_messages: [Broadway.Message.t],
  failed_messages: [Broadway.Message.t],
  batch_info: Broadway.BatchInfo.t,
  telemetry_span_context: reference,
  producer: {atom, list}
}



	[:broadway, :batcher, :start] - Dispatched by a Broadway batcher before
handling events
	Measurement: %{system_time: integer}

	Metadata:
%{
  topology_name: atom,
  name: atom,
  batcher_key: atom,
  messages: [Broadway.Message.t],
  telemetry_span_context: reference
}



	[:broadway, :batcher, :stop] - Dispatched by a Broadway batcher after
handling events
	Measurement: %{duration: native_time}
	Metadata:

  %{
    topology_name: atom,
    name: atom,
    batcher_key: atom,
    telemetry_span_context: reference
  }


Most of the events follow the :telemetry.span/3 convention for measurements.
This means that "start" events have a :system_time representing the start of
that event using System.system_time/0. The "stop" or "exception" events
have the duration value, which is the difference in monotonic time between
the start and stop events.

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Broadway.BatchInfo - Broadway v1.1.0
    
    

    


  
  

    
Broadway.BatchInfo 
    



      
A struct used to hold information about a generated batch.
An instance of this struct containing the related info will
be passed to the Broadway.handle_batch/4 callback of the
module implementing the Broadway behaviour.
See the documentation for %Broadway.BatchInfo{}
for information on the fields.

    Broadway.Message - Broadway v1.1.0
    
    

    


  
  

    
Broadway.Message 
    



      
This struct holds all information about a message.
A message is first created by the producers. It is then
sent downstream and gets updated multiple times, either
by a module implementing the Broadway behaviour
through the Broadway.handle_message/3 callback
or internally by one of the built-in stages of Broadway.
Instead of modifying the struct directly, you should use the functions
provided by this module to manipulate messages. However, if you are implementing
a Broadway.Producer of your own, see the %Broadway.Message{}
documentation to see what fields you should set.

Broadway.Acknowledger behaviour
    



      
A behaviour used to acknowledge that the received messages
were successfully processed or failed.
When implementing a new connector for Broadway, you should
implement this behaviour and consider how the technology
you're working with handles message acknowledgement.
The ack/3 callback must be implemented in order to notify
the origin of the data that a message can be safely removed
after been successfully processed and published. In case of
failed messages or messages without acknowledgement, depending
on the technology chosen, the messages can be either moved back
in the queue or, alternatively, moved to a dead-letter queue.

    Broadway.CallerAcknowledger - Broadway v1.1.0
    
    

    


  
  

    
Broadway.CallerAcknowledger 
    



      
A simple acknowledger that sends a message back to a caller.
If you want to use this acknowledger in messages produced by your
Broadway.Producer, you can get its configuration by calling
the init/0 function. For example, you can use it in
Broadway.test_message/3:
some_ref = make_ref()

Broadway.test_message(
  MyPipeline,
  "some data",
  acknowledger: Broadway.CallerAcknowledger.init({self(), some_ref}, :ignored)
)
The first parameter is a tuple with the PID to receive the messages
and a unique identifier (usually a reference). Such unique identifier
is then included in the messages sent to the PID. The second parameter,
which is per message, is ignored.
It sends a message in the format:
{:ack, ref, successful_messages, failed_messages}
If Broadway.Message.configure_ack/2 is called on a message that
uses this acknowledger, then the following message is sent:
{:configure, ref, options}

Broadway.NoopAcknowledger 
    



      
An acknowledger that does nothing.
If you want to use this acknowledger in messages produced by your
Broadway.Producer, you can get its configuration by calling
the init/0 function. For example, you can use it in
Broadway.test_message/3:
Broadway.test_message(MyPipeline, "some data", acknowledger: Broadway.NoopAcknowledger.init())
Broadway sets this acknowledger automatically on messages that have been acked
via Broadway.Message.ack_immediately/1.

Broadway.DummyProducer 
    



      
A producer that does nothing, used mostly for testing.
See "Testing" section in Broadway module documentation for more information.

Broadway.Producer behaviour
    



      
A Broadway producer is a GenStage producer that emits
Broadway.Message structs as events.
The Broadway.Producer is declared in a Broadway topology
via the :module option (see Broadway.start_link/2):
producer: [
  module: {MyProducer, options}
]
Once declared, MyProducer is expected to implement and
behave as a GenStage producer. When Broadway starts,
the GenStage.init/1 callback will be invoked directly with the
given options.

  
    
  
  Injected Broadway configuration


If options is a keyword list, Broadway injects a :broadway option
into such keyword list. This option contains the configuration for the
complete Broadway topology (see Broadway.start_link/2. For example, you can use
options[:broadway][:name] to uniquely identify the topology,
allowing you to write terms to things such as
:persistent_term
or ETS tables.
The :broadway configuration also has an :index key. This
is the index of the producer in its supervision tree (starting
from 0). This allows a features such having even producers
connect to some server while odd producers connect to another.
If options is any other term, it is passed as is to the GenStage.init/1
callback. All other functions behave precisely as in GenStage
with the requirements that all emitted events must be Broadway.Message
structs.

  
    
  
  Optional callbacks


A Broadway.Producer can implement two optional Broadway callbacks,
prepare_for_start/2 and prepare_for_draining/1, which are useful
for booting up and shutting down Broadway topologies respectively.

  
    
  
  Producing Broadway messages


You should generally modify Broadway.Message structs by using the functions
in the Broadway.Message module. However, if you are implementing your
own producer, you can manipulate some of the struct's fields directly.
These fields are:
	:data (required) - the data of the message. Even though the function
Broadway.Message.put_data/2 exists, when creating a %Broadway.Message{}
struct from scratch you will have to pass in the :data field directly.

	:acknowledger (required) - the acknowledger of the message, of type
Broadway.Message.acknowledger/0.

	:metadata (optional) - metadata about the message that your producer
can attach to the message. This is useful when you want to add some metadata
to messages, and document it for users to use in their pipelines.


For example, a producer could create a message by doing something like this:
%Broadway.Message{
  data: "some data here",
  acknowledger: Broadway.NoopAcknowledger.init()
}

